SG Annotation

SAS provides a general way to customize plots with text, images, and symbols by using SG Annotation. This method is used with the sg procedures by indicating a name of a separate data set with a `sganno` option. The data set does not contain the main data being plotted, but rather the desired set of annotations with their locations relative to the plot.

There is one new program and one new data file for these notes: cereal_graphics_sganno.sas and sganno_values.csv.

Labeling points

The sgplot procedure can label all or some observations using the `datalabel` option.

```
dm "log; clear; odsresults; clear;";

title1 "Chris Bilder, STAT 850";

proc import out=cereal datafile="C:\data\cereal.csv"
                DBMS=CSV replace;
  getnames=yes;
  datarow=2;
run;

*Adjust for serving size;
data set1;
  set cereal;
  sugar = sugar_g/size_g;
  fat = fat_g/size_g;
  sodium = sodium_mg/size_g;

  *remove the old variables below from the data set;
  drop size_g sugar_g fat_g sodium_mg;
run;
```

```
/*
title2 "Cereal data adjusted for serving size";
proc print data=set1(obs = 20);
run;
*/

*attrpriority=none allows one to control styles better;
ods graphics / attrpriority=none;

title2 "Sugar vs. fat with observation numbers";
proc sgplot data=set1;
  StyleAttrs datasymbols=(circle diamond square triangle)
    datacontrastcolors=(red blue green purple);
  scatter x=fat y=sugar / group=shelf MarkerAttrs=(size=10)
    datalabel=ID;
  yaxis grid GridAttrs=(pattern=dot color=gray);
  xaxis grid GridAttrs=(pattern=dot color=gray);
run;
```




```
* Label just one observation;
```

```
data set1;
  set set1;
  if ID = 20 then ID_oreo = "Oreo's";  * Discussed in datastep
    section;
run;

title2 "Suagr vs. fat with just observation #20 labeled";
proc sgplot data=set1;
  StyleAttrs datasymbols=(circle diamond square triangle)
    datacontrastcolors=(red blue green purple);
  scatter x=fat y=sugar / group=shelf MarkerAttrs=(size=10)
    datalabel=ID_oreo;
  yaxis grid GridAttrs=(pattern=dot color=gray);
  xaxis grid GridAttrs=(pattern=dot color=gray);
run;
```



A more flexible way is to use SG annotation. The code next shows how to label only the Oreo's observation.

```
* This creates a data set with 1 observation to use with SG
  Annotation;
```

```
data sganno1;
  function = "text"; * Using text for annotation;
  anchor = "center"; * Place to put the text based on the x and
    y values (this is default);
  label = "Oreo's"; * The actual text;
  x1space = "datavalue"; * Position text based on x-axis
    coordinates rather than the default of % on screen;
  y1space = "datavalue"; * Position text based on y-axis
    coordinates rather than the default of % on screen;
  x1 = 0.085; * Location based on x-axis;
  y1 = 0.41;  * Location based on y-axis;
  width = 0.02; * Available width of the text field on the plot;
  widthunit = "data";  * Based on the x-axis coordinates,
    default is percent;
  textsize = 14;  * Font size;
run;

title2 "Example #1 of SG Annotation";
proc sgplot data=set1 sganno=sganno1;
  StyleAttrs datasymbols=(circle diamond square triangle)
    datacontrastcolors=(red blue green purple);
  scatter x=fat y=sugar / group=shelf MarkerAttrs = (size=10);
  yaxis grid GridAttrs=(pattern=dot color=gray);
  xaxis grid GridAttrs=(pattern=dot color=gray);
run;
```




```
* Create same plot but now with an arrow as well;

* This creates a data set with 1 observation to use with SG
   Annotation;
data sganno1;
  length function $ 10;  * Without specifying the length of the
    text, "arrow" in next sganno gets cut off;
  function = "text";
  anchor = "right";
  label = "Oreo's";
  x1space = "datavalue";
  y1space = "datavalue";
  x1 = 0.085;
  y1 = 0.41;
  width = 0.02;
  widthunit = "data";
  textsize = 14;
run;
```

```
data sganno2;
  length function $ 10;  * Without specifying the length of the
    text, "arrow" gets cut off;
  function = "arrow";
  linecolor = "black";
  shape = "barbed";
  direction = "out"; * Want arrow to point to (x2, y2);
  drawspace = "datavalue";
  x1 = 0.085;  * Location based on x-axis;
  y1 = 0.41;   * Location based on y-axis;
  x2 = 0.091;  * End x-axis location of arrow;
  y2 = 0.41;   * End y-axis location of arrow;
run;

*Combine the annotations;
data sganno_all;
  set sganno1 sganno2;
run;

title2 "sganno data set";
proc print data=sganno_all;
run;
```

Chris Bilder, STAT 850
sganno data set

Obs	function	anchor	label	x1space	y1space	x1	y1	width	widthunit	textsize	linecolor	shape	direction	drawspace	x2	y2
1	text	right	Oreo's	datavalue	datavalue	0.085	0.41	0.02	data	14					.	.
2	arrow					0.085	0.41	.		.	black	barbed	out	datavalue	0.091	0.41

```
title2 "Illustrate #1 of SG Annotation";
proc sgplot data=set1 sganno=sganno_all;
  StyleAttrs datasymbols=(circle diamond square triangle)
    datacontrastcolors=(red blue green purple);
  scatter x=fat y=sugar / group=shelf MarkerAttrs = (size=10);
  yaxis grid GridAttrs=(pattern=dot color=gray);
  xaxis grid GridAttrs=(pattern=dot color=gray);
run;
```



There are other ways to create the data set for the `sganno` option (see program):

- Create an Excel file and read the file into SAS.
- Use a `datalines` statement in a data step so that the two rows of sganno_all are created at the same time.
- Use `%sgtext` and `%sgarrow` macros within data steps. Macros are discussed later in the course.

Additional items

- There are many other items that may be of interest to add to plots, such as lines. The function statement in the annotation statement is the key to specify the type of item. For example, to add a line, use `function = "line"`.
- Where to find help?



- There is a similar annotation method for the g procedures as well.